// File: contracts/TicketStaking.sol



pragma solidity ^0.8.15;

import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC721/utils/ERC721Holder.sol";

import "./Interfaces/ITickets.sol";
import "./Interfaces/IMaestro.sol";

/**
 * https://opus-labs.io
 * @title   Maestro - Tickets Staking
 * @notice  Stake Maestro - Tickets NFTs
 * @author  BowTiedPickle
 */
contract TicketStaking is Ownable, ReentrancyGuard, ERC721Holder {
    event Stake(uint256 indexed token, address indexed depositor);
    event Unstake(uint256 indexed token, address indexed depositor);
    event StakingEnabled(bool newStatus);
    event TierWeightUpdated(uint8 tier, uint256 weight);

    ITickets public immutable tickets;
    IMaestro public maestro;

    struct StakeInfo {
        address depositor;
        uint8 tier;
        uint256 stakeTime;
    }

    mapping(uint256 => StakeInfo) public tokenToStakeInfo;

    bool public stakingEnabled;
    uint256 public totalStaked;

    uint8 public constant TICKET_DIAMOND = 1;
    uint8 public constant TICKET_GOLD = 2;
    uint8 public constant TICKET_SILVER = 3;
    uint8 public constant TICKET_BRONZE = 4;

    mapping(uint8 => uint256) public tierToWeight;

    // ----- Construction -----

    /**
     * @param   _tickets    Address of the Maestro - Tickets contract
     */
    constructor(ITickets _tickets) {
        tickets = _tickets;
        tierToWeight[TICKET_DIAMOND] = 20_000;
        tierToWeight[TICKET_GOLD] = 15_000;
        tierToWeight[TICKET_SILVER] = 12_500;
        tierToWeight[TICKET_BRONZE] = 10_000;
    }

    // ----- Public Functions -----

    /**
     * @notice  Stake a Maestro - Ticket NFT
     * @param   _token  Token ID of the ticket to stake
     */
    function stake(uint256 _token) external nonReentrant {
        require(stakingEnabled, "!enabled");
        stakeInternal(_token);
    }

    /**
     * @notice  Stake multiple Maestro - Ticket NFTs
     * @param   _tokens   Token IDs of the tickets to stake
     */
    function stakeMultiple(uint256[] calldata _tokens) external nonReentrant {
        require(stakingEnabled, "!enabled");
        uint256 len = _tokens.length;
        for (uint256 i; i < len; ) {
            stakeInternal(_tokens[i]);
            unchecked {
                ++i;
            }
        }
    }

    /**
     * @notice  Unstake a Maestro - Ticket NFT
     * @param   _token  Token ID of the ticket to unstake
     */
    function unstake(uint256 _token) external nonReentrant {
        unstakeInternal(msg.sender, _token);
    }

    /**
     * @notice  Unstake multiple Maestro - Ticket NFTs
     * @param   _tokens   Token IDs of the tickets to unstake
     */
    function unstakeMultiple(uint256[] calldata _tokens) external nonReentrant {
        uint256 len = _tokens.length;
        for (uint256 i; i < len; ) {
            unstakeInternal(msg.sender, _tokens[i]);
            unchecked {
                ++i;
            }
        }
    }

    function stakeInternal(uint256 _token) internal {
        StakeInfo storage stakeInfo = tokenToStakeInfo[_token];
        require(stakeInfo.depositor == address(0), "!staked");

        ++totalStaked;
        stakeInfo.depositor = msg.sender;
        stakeInfo.stakeTime = block.timestamp;
        uint8 tier = tickets.ticketTier(_token);
        stakeInfo.tier = tier;

        maestro.notifyTicketStaked(true, msg.sender, tier);

        // Intake token
        tickets.safeTransferFrom(msg.sender, address(this), _token);

        emit Stake(_token, msg.sender);
    }

    function unstakeInternal(address _to, uint256 _token) internal {
        StakeInfo storage stakeInfo = tokenToStakeInfo[_token];
        require(stakeInfo.depositor == _to, "!owner");

        --totalStaked;
        stakeInfo.depositor = address(0);
        stakeInfo.stakeTime = 0;
        uint8 tier = stakeInfo.tier;
        stakeInfo.tier = 0;

        maestro.notifyTicketStaked(false, _to, tier);

        // Send out token
        tickets.safeTransferFrom(address(this), _to, _token);

        emit Unstake(_token, _to);
    }

    // ----- View Functions -----

    /**
     * @notice  View the time a ticket has been staking
     * @param   _token  Token ID of the ticket
     */
    function timeStaking(uint256 _token) external view returns (uint256) {
        uint256 stakeTime = tokenToStakeInfo[_token].stakeTime;
        return stakeTime > 0 ? block.timestamp - stakeTime : 0;
    }

    /**
     * @notice  View the Notes a ticket has accumulated
     * @param   _token  Token ID of the ticket
     */
    function notesBalance(uint256 _token) external view returns (uint256) {
        StakeInfo memory stakeInfo = tokenToStakeInfo[_token];
        uint256 stakeTime = stakeInfo.stakeTime;
        uint256 tierWeight = tierToWeight[stakeInfo.tier];
        if (stakeTime > 0) {
            return ((block.timestamp - stakeTime) * tierWeight) / 10_000;
        } else {
            return 0;
        }
    }

    /**
     * @notice  View if a ticket is currently staked
     * @param   _token  Token ID of the ticket
     */
    function isStaked(uint256 _token) external view returns (bool) {
        return tokenToStakeInfo[_token].stakeTime > 0;
    }

    /**
     * @notice  View all staking information for a ticket
     * @param   _token  Token ID of the ticket
     */
    function getStakeInfo(uint256 _token)
        external
        view
        returns (StakeInfo memory)
    {
        return tokenToStakeInfo[_token];
    }

    // ----- Admin Functions -----

    /**
     * @notice  Force unstake a user in case staking needs to be disabled
     * @dev     Permissioned
     * @param   _token  Token ID of the ticket to unstake
     */
    function unstakeFor(address _to, uint256 _token)
        external
        nonReentrant
        onlyOwner
    {
        unstakeInternal(_to, _token);
    }

    /**
     * @notice  Enable or disable staking
     * @param   _status     True for enabled, false for disabled
     */
    function setStakingEnabled(bool _status) external onlyOwner {
        stakingEnabled = _status;
        emit StakingEnabled(_status);
    }

    /**
     * @notice  Set the Notes generation weight of a ticket tier
     * @dev     Set to 0 to disable Notes for this ticket
     * @param   _tier   Ticket tier
     * @param   _weight New weight, where <10,000 is a penalty, 10,000 is neutral, and greater than 10,000 is a boost
     */
    function setTierWeight(uint8 _tier, uint256 _weight) external onlyOwner {
        require(_tier <= TICKET_BRONZE && _tier > 0, "!tier");
        tierToWeight[_tier] = _weight;
        emit TierWeightUpdated(_tier, _weight);
    }

    // ----- Irreversible Admin Functions -----

    /**
     * @notice  Irreversibly set the Maestro - Genesis contract address
     * @dev     May only be called once
     * @param   _maestro    Address of Maestro - Genesis contract
     */
    function setMaestro(IMaestro _maestro) external onlyOwner {
        require(address(maestro) == address(0), "!initialized");
        maestro = _maestro;
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC721/utils/ERC721Holder.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/utils/ERC721Holder.sol)

pragma solidity ^0.8.0;

import "../IERC721Receiver.sol";

/**
 * @dev Implementation of the {IERC721Receiver} interface.
 *
 * Accepts all token transfers.
 * Make sure the contract is able to use its token with {IERC721-safeTransferFrom}, {IERC721-approve} or {IERC721-setApprovalForAll}.
 */
contract ERC721Holder is IERC721Receiver {
    /**
     * @dev See {IERC721Receiver-onERC721Received}.
     *
     * Always returns `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address,
        address,
        uint256,
        bytes memory
    ) public virtual override returns (bytes4) {
        return this.onERC721Received.selector;
    }
}

// File: contracts/Interfaces/ITickets.sol



pragma solidity ^0.8.15;

import "@openzeppelin/contracts/token/ERC721/IERC721.sol";

interface ITickets is IERC721 {
    /**
     * @notice  Returns the tier of the given token
     * @param   tokenID    TokenID to query
     * @return  0 if token nonexistent, 1 through 4 otherwise
     */
    function ticketTier(uint256 tokenID) external view returns (uint8);
}

// File: contracts/Interfaces/IMaestro.sol



pragma solidity ^0.8.15;

import "@openzeppelin/contracts/token/ERC721/IERC721.sol";

interface IMaestro is IERC721 {
    function notifyTicketStaked(
        bool _status,
        address _user,
        uint8 _tier
    ) external;
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}